

Optimisation géométrique de forme sous FreeFem++

O. Pantz
CMAP
Ecole Polytechnique

September 22, 2005

Position du problème

Optimisation de forme

Déterminer la structure optimale Ω^* minimisant une fonction coût J .

$$J(\Omega^*) = \min_{\Omega} J(\Omega).$$

Pourquoi FreeFem++ ?

- Gestion des maillages

La forme Ω est représentée par un maillage.

- Résolution aisée d'EDP

$J(\Omega)$ dépend de la solution d'une EDP.

La méthode du gradient

Soit $J : V \rightarrow \mathbb{R}$. La méthode du gradient consiste à construire une suite x_n définie par

$$x_{n+1} = x_n - h_n d_n,$$

où

$$(d_n, y)_V = \langle J'(x_n), y \rangle.$$

Problème

L'ensemble des ouverts de $\mathbb{R}^2 \neq$ espace de Hilbert !

Structure différentielle ?

Définition de J' ?

Quel produit scalaire ?

Variations de forme

Soit $\theta \in W^{1,\infty}(\mathbb{R}^2)^2$. On pose $\Omega_\theta = (\text{Id} + \theta)(\Omega)$.

Dérivation de forme

On dit que J est dérivable en Ω si l'application $\theta \in W^{1,\infty}(\mathbb{R}^2)^2 \mapsto J(\Omega_\theta)$ est dérivable en $\theta = 0$.

Produit scalaire de l'algorithme de gradient

L'espace $W^{1,\infty}(\mathbb{R}^2)^2$ n'est pas un Hilbert.

On choisit $\theta \in V = H^1(\mathbb{R}^2)^2$.

Moralement, il faudrait choisir $H^2(\mathbb{R}^2)^2$,
mais $H^1(\mathbb{R}^2)^2$ marche très bien d'un point de vue pratique...

Méthode de gradient en optimisation de forme

1. Choisir Ω_0 , $n = 0$.
2. Déterminer $d_n \in V$ solution du problème variationnel

$$(d_n, \theta)_V = \langle J'(\Omega_n), \theta \rangle$$

pour tout $\theta \in V$ (où $V = (H^1(\Omega))^2$).

3. On transporte la forme le long du champ d_n

$$\Omega_{n+1} = (\text{Id} - h_n d_n)(\Omega_n).$$

4. Si $d_n = 0$, on arrête. Sinon, on incrémente n et on retourne à l'étape 2.

Minimisation de la compliance en élasticité

Le déplacement

$$u \in W = \{v \in H^1(\Omega)^2 : v = 0 \text{ sur } \Gamma_D\}$$

est solution du problème variationnel

$$\int_{\Omega} 2\mu e(u) : e(v) + \lambda(\operatorname{div} u)(\operatorname{div} v) dx = \int_{\Gamma_N} g \cdot v ds.$$


La compliance est définie par

$$J(\Omega) = \int_{\Gamma_N} g \cdot u ds$$

et

$$\langle J'(\Omega), \theta \rangle = - \int_{\Gamma_{opt}} (2\mu |e(u)|^2 + \lambda(\operatorname{div} u)^2) \theta \cdot n dx$$

Forme Initiale



Détails techniques

1. On ajoute une contrainte sur le volume. A cet effet, on introduit le multiplicateur de Lagrange associé qui est renouvelé à chaque itération.
2. On déplace les noeuds du maillage à chaque itérations. De plus, on adapte fréquemment le maillage afin qu'il ne dégénère pas.
3. Lors du déplacement des noeuds, on vérifie qu'aucun triangle n'est retourné.
4. On adapte le pas de descente h_n de sorte à optimiser la convergence.
5. On utilise deux maillages afin de d'éviter les oscillations: Un grossier décrivant le domaine Ω_n l'autre fin pour calculer $u(\Omega_n)$ et d_n .

Définition du maillage initial

```
int neumann=1; int dirichlet=2; int free=3;

border a(t=-1,1) {x=20; y=t;label=neumann;};
border b(t=1,0) {x=20.*t;y=4.-3.*t;label=free;};
border c1(t=4,2) {x=0; y=t;label=dirichlet;};
border c2(t=2,-2) {x=0; y=t;label=free;};
border c3(t=-2,-4){x=0; y=t;label=dirichlet;};
border d(t=0,1) {x=20.*t; y=-4.+3.*t;label=free;};

border Hole1(t=0,2*pi)
    {x=0.2*cos(t)+8;y=0.2*sin(t);label=free;};

mesh Sh = buildmesh (a(20)+b(30)+c1(10)+c2(10)
    +c3(10)+d(30)+Hole1(-11));
```

Elasticité

```
real E=15; //Young Modulus
real nu=0.35; //Poisson ratio
real lambda=E*nu/((1.+nu)*(1.-2.*nu));
real mu=E/(2.*(1.+nu));
func g1=0.; func g2=-1.;

fespace WSh(Sh, [P2,P2]);
WSh [u1,u2],[v1,v2];
problem elasticity([u1,u2],[v1,v2]) =
  int2d(Sh)(
    2.*mu*(dx(u1)*dx(v1)+dy(u2)*dy(v2)
    +((dx(u2)+dy(u1))*(dx(v2)+dy(v1))))/2.)
    +lambda*(dx(u1)+dy(u2))*(dx(v1)+dy(v2)))
  -int1d(Sh,neumann)(g1*v1+g2*v2)
  +on(dirichlet,u1=0,u2=0);
```


Direction de descente

```
real lagrange;
macro gradientexp()
  -2.*mu*(dx(u1)^2+dy(u2)^2
  +((dx(u2)+dy(u1))^2)/2.)
  -lambda*(dx(u1)+dy(u2))^2//
WSh [d1,d2];
WSh [theta1,theta2];
macro prodscal(t1,t2,p1,p2)
  dx(t1)*dx(p1)+dy(t1)*dy(p1)
  +dx(t2)*dx(p2)+dy(t2)*dy(p2)+t1*p1+t2*p2//
problem extension([d1,d2],[theta1,theta2]) =
  int2d(Sh)(prodscal(d1,d2,theta1,theta2))
  +int1d(Sh,free)( (theta1*N.x+theta2*N.y)
    *(gradientexp + lagrange))
  +on(dirichlet,neumann,d1=0,d2=0);
```

Boucle d'optimisation


```
for (iter=0;iter< niter;iter=iter+1){  
    elasticity;  
  
    volume=int1d(Sh) (x*N.x+y*N.y)/2;  
    perimeter=int1d(Sh,free) (1.);  
    lagrange=0.5*lagrange  
        +0.5*int1d(Sh,free) (gradientexp)/perimeter  
        +lagrangestep*(volume-volume0)/volume0;  
  
    extension;  
  
    Sh = movemesh (Sh, [x+step*d1,y+step*d2]);  
    plot(Sh);  
};
```

Résultat Numériques




Compliance=23.3

Résultat Numériques




Compliance=19.6

Résultat Numériques




Compliance=18.3

Résultat Numériques




Compliance=17.6

Résultat Numériques




Compliance=17.5

Résultat Numériques



Compliance=17.4

Résultat Numériques



Compliance=17.1